
Introduction
ExUssd lets you create simple, flexible, and customizable USSD interface.
Under the hood ExUssd uses Elixir Registry to create and route individual USSD session.

Installation
available in Hex, the package can be installed
by adding ex_ussd to your list of dependencies in mix.exs:
def deps do
  [
    {:ex_ussd, "0.1.6"}
  ]
end
Example
Checkout The example folder.
Configuration
The ExUssd field can be set using Use ExUssd.set/2
Following are the setable fields
@allowed_fields [
    :title, # -> USSD menu title
    :next, # -> %{name: "MORE", next: "98", delimiter: ":"}
    :previous, # -> %{name: "BACK", previous: "0", delimiter: ":"}
    :home, # -> %{name: "HOME", previous: "00", delimiter: ":"}
    :should_close, # -> Indicate Menu state, default `false`
    :split, # -> Set split menu list by,  default 7
    :delimiter, # -> Set delimiter style,  ":"
    :error, # -> Custom error message, Used in the `callback/2`
    :default_error, # -> Set default error message, "Invalid Choice\n", Used in the `init/2`
    :show_navigation # Show navigation, default `true`  
    :data
  ]
Set ExUssd Global Default config.exs
  config :ex_ussd,
    default: [
        next: %{name: "MORE", next: "98", delimiter: ":"},
        previous: %{name: "BACK", previous: "0", delimiter: ":"},
        split: 7,
        default_error: "You have selected invalid option try again\n",
        delimiter: ":"
    ]
Simple USSD menu
Implement ExUssd init/2 or init/3 callback.
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)

  api_parameters = %{"service_code" => "*544#", "session_id" => "session_01", "text" => ""}

  ExUssd.goto(menu: menu, api_parameters: api_parameters)

  {:ok, %{menu_string: "Welcome", should_close: false}}
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters, _metadata) do
      menu 
      |> ExUssd.set(title: "Welcome")
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)

  api_parameters = %{"service_code" => "*544#", "session_id" => "session_01", "text" => ""}

  ExUssd.goto(menu: menu, api_parameters: api_parameters)

  {:ok, %{menu_string: "Welcome", should_close: false}}
End USSD Session
Manually close USSD session, Use ExUssd.end_session/1 it takes the session_id as params
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.set(should_close: true)
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)

  api_parameters = %{"service_code" => "*544#", "session_id" => "session_01", "text" => ""}

  ExUssd.goto(menu: menu, api_parameters: api_parameters)
    |> case do
    {:ok, %{menu_string: menu_string, should_close: false}} ->
      "CON " <> menu_string

    {:ok, %{menu_string: menu_string, should_close: true}} ->
      # End Session
      ExUssd.end_session(session_id: "session_01")

      "END " <> menu_string
    end
USSD Simple List
Use ExUssd.add/2 to add to USSD menu list.
The USSD menu list is [] by default.
  defmodule ProductAHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu |> ExUssd.set(title: "selected product a")
    end
  end

  defmodule ProductBHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu |> ExUssd.set(title: "selected product b")
    end
  end

  defmodule ProductCHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "selected product c")
    end
  end
  
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.add(ExUssd.new(name: "Product A", handler: ProductAHandler))
      |> ExUssd.add(ExUssd.new(name: "Product B", handler: ProductBHandler))
      |> ExUssd.add(ExUssd.new(name: "Product C", handler: ProductCHandler))
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)
USSD Nested List
Use ExUssd.add/2 to add to USSD menu list on Individual USSD menu.
  
  defmodule ProductCHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "selected product c")
    end
  end
  
  defmodule ProductBHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "selected product b")
      |> ExUssd.add(ExUssd.new(name: "Product C", handler: ProductCHandler))
    end
  end

  defmodule ProductAHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "selected product a")
      |> ExUssd.add(ExUssd.new(name: "Product B", handler: ProductBHandler))
    end
  end

  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.add(ExUssd.new(name: "Product A", handler: ProductAHandler))    
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)
Using USSD after_route/1
Implement after_route/1 function on your USSD handler module.
after_route/1 callback returns navigation status.
Scenario
User passes in invalid
	response %{
state: :error,
menu: menu,
payload: %{
  api_parameters: api_parameters,
  metadata: metadata
}
}


User passes in valid input, name navigated to next menu
	response %{
 state: :ok,
 payload: %{
   api_parameters: api_parameters,
   metadata: metadata
 }
}


  # ...
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.add(ExUssd.new(name: "Product A", handler: ProductAHandler))
      |> ExUssd.add(ExUssd.new(name: "Product B", handler: ProductBHandler))
      |> ExUssd.add(ExUssd.new(name: "Product C", handler: ProductCHandler))
    end

    def after_route(%{state: :ok, payload: payload}) do
      IO.inspect payload
    end

    def after_route(%{state: :error, menu: menu, payload: payload}) do
      IO.inspect payload
      # menu |> ExUssd.set(title: "Max retry reached, account locked") |> ExUssd.set(should_close: true)
    end
  end
  
  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)
Using USSD callback
Implement ExUssd callback/2 or callback/3 in the event you need to validate the Users input 
Simple validation menu
  # ...
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.add(ExUssd.new(name: "Product A", handler: ProductAHandler))
      |> ExUssd.add(ExUssd.new(name: "Product B", handler: ProductBHandler))
      |> ExUssd.add(ExUssd.new(name: "Product C", handler: ProductCHandler))
    end

    def callback(menu, api_parameters) do
      case api_parameters.text == "5555" do
        true ->
          menu
          |> ExUssd.set(title: "You have Entered the Secret Number, 5555")
          |> ExUssd.set(should_close: true)

        _ ->
          menu
      end
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)

  api_parameters = %{"service_code" => "*544#", "session_id" => "session_01", "text" => "5555"}

  ExUssd.goto(menu: menu, api_parameters: api_parameters)

  {:ok, %{menu_string: "You have Entered the Secret Number, 5555", should_close: false}}
Nested validation menu
  # ...
  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(%{data: data} = menu, _api_parameters) do
      IO.inspect data
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.add(ExUssd.new(name: "Product A", handler: ProductAHandler))
      |> ExUssd.add(ExUssd.new(name: "Product B", handler: ProductBHandler))
      |> ExUssd.add(ExUssd.new(name: "Product C", handler: ProductCHandler))
    end

    def callback(menu, api_parameters) do
      case api_parameters.text == "5555" do
        true ->
          menu
          |> ExUssd.set(title: "You have Entered the Secret Number, 5555")
          |> ExUssd.set(should_close: true)

        _ ->
          menu
      end
    end
  end

  defmodule PinHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Enter your pin number")
      |> ExUssd.set(show_navigation: false)
    end

    def callback(menu, api_parameters) do
      case api_parameters.text == "4321" do
        true ->
          menu
          # |> ExUssd.navigate(data: %{name: "John"}, handler: MyHomeHandler)
          |> ExUssd.navigate(handler: MyHomeHandler)
        _ ->
          menu 
          |> ExUssd.set(error: "Wrong pin number\n")
      end
    end

    def after_route(payload) do
      IO.inspect payload
    end
  end

  menu = ExUssd.new(name: "Check PIN", handler: PinHandler)
  # ...
Using USSD dynamic
Dymanic Vertical menus
  # ...
  defmodule SubCountyHandler do
    use ExUssd.Handler
    def init(%{data: %{name: name}} = menu, api_parameters) do
      # TODO: Fetch county sub locations by county_code
      # Make dynamic location menus for the county
      # Split by 6 / 7
      menu 
      |> ExUssd.set(title: "#{name} County")
    end
  end

  defmodule CountyHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menus =
        fetch_api()
        |> Enum.map(fn %{name: name} = data ->
          ExUssd.new(name: name, data: data)
        end)

      menu
      |> ExUssd.set(title: "List of Counties")
      |> ExUssd.dynamic(
        menus: menus,
        handler: App.Dymanic.Vertical.SubCountyHandler,
        orientation: :vertical
      )
    end

    def fetch_api do
      [
        %{county_code: 47, name: "Nairobi"},
        %{county_code: 01, name: "Mombasa"},
        %{county_code: 42, name: "Kisumu"}
      ]
    end
  end

  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "Welcome")
      |> ExUssd.add(ExUssd.new(name: "Counties List", handler: CountyHandler))
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)
  # ...
  {:ok, %{
   menu_string: "List of Counties\n1:Nairobi\n2:Mombasa\n3:Kisumu\n0:BACK",
   should_close: false
 }}
Dymanic Horizontal menus
Note: The name value is Truncated after 140 characters

  defmodule NewsHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menus = fetch_api() |> Enum.map(fn %{"title"=> title, "body"=> body} -> 
           ExUssd.new(name: title <> "\n" <> body)
      end)

      menu 
      |> ExUssd.set(title: "World News")
      |> ExUssd.dynamic(menus: menus, orientation: :horizontal)
    end

    def fetch_api do
    [
      %{
        "userId" => 1,
        "id" => 1,
        "title" => "sunt aut facere repellat provident occaecati excepturi optio reprehenderit",
        "body" =>
          "quia et suscipit suscipit recusandae consequuntur expedita et cum reprehenderit molestiae ut ut quas totam nostrum rerum est autem sunt rem eveniet architecto"
      },
      %{
        "userId" => 1,
        "id" => 2,
        "title" => "qui est esse",
        "body" =>
          "est rerum tempore vitae sequi sint nihil reprehenderit dolor beatae ea dolores neque fugiat blanditiis voluptate porro vel nihil molestiae ut reiciendis qui aperiam non debitis possimus qui neque nisi nulla"
      },
      %{
        "userId" => 1,
        "id" => 3,
        "title" => "ea molestias quasi exercitationem repellat qui ipsa sit aut",
        "body" =>
          "et iusto sed quo iure voluptatem occaecati omnis eligendi aut ad voluptatem doloribus vel accusantium quis pariatur molestiae porro eius odio et labore et velit aut"
      }
    ]
    end
  end

  defmodule MyHomeHandler do
    use ExUssd.Handler
    def init(menu, _api_parameters) do
      menu 
      |> ExUssd.set(title: "BBC News")
      |> ExUssd.add(ExUssd.new(name: "News", handler: NewsHandler))
      # |> ExUssd.add(ExUssd.new(name: "WorkLife", handler: WorkLifeHandler))
      # |> ExUssd.add(ExUssd.new(name: "Sports", handler: SportsHandler))
    end
  end

  menu = ExUssd.new(name: "Home", handler: MyHomeHandler)
  # ...
  {:ok, %{
   menu_string: "1/3\nsunt aut facere repellat provident occaecati excepturi optio reprehenderit\nquia et suscipit suscipit recusandae consequuntur expedita et ...\n0:BACK 98:MORE",
   should_close: false
 }}
Phoenix Simulator
Update your router's configuration to forward requests to ExUssd Simlutate with a menu entry and phone_numbers list
# lib/my_app_web/router.ex
use MyAppWeb, :router

import ExUssd
...

if Mix.env() == :dev do
  scope "/" do
    pipe_through :browser
    simulate "/simulator",
      menu: ExUssd.new(name: "Home", handler: MyHomeHandler),
      phone_numbers: ["254700100100", "254700200200", "254700300300"]
  end
end



  

Guide
If you don't see your idea listed, and you think it fits into the goals of this library, do the following:
	Find an issue that you are interested in addressing or a feature that you would like to add.
	Fork the repository associated with the issue to your local GitHub organization. This means that you will have a copy of the repository under your-GitHub-username/repository-name.
	Clone the repository to your local machine using git clone https://github.com/beamkenya/ex_ussd.git
	Create a new branch for your fix using git checkout -b your-branch-name-here.
	Make the appropriate changes for the issue you are trying to address or the feature that you want to add.
	Use git add insert-paths-of-changed-files-here to add the file contents of the changed files to the "snapshot" git uses to manage the state of the project, also known as the index.
	Use git commit -m "Insert a short message of the changes made here" to store the contents of the index with a descriptive message.
	Push the changes to the remote repository using git push origin your-branch-name-here.
	Submit a pull request to the upstream repository.
	Title the pull request with a short description of the changes made and the issue or bug number associated with your change. For example, you can title an issue like so "Added more log outputting to resolve #4352".
	In the description of the pull request, explain the changes that you made, any issues you think exist with the pull request you made, and any questions you have for the maintainers. It's OK if your pull request is not perfect (no pull request is), the reviewer will be able to help you fix any problems and improve it!
	Wait for the pull request to be reviewed by a maintainer.
	Make changes to the pull request if the reviewing maintainer recommends them.
	Celebrate your success after your pull request is merged!

Testing
Make sure you write tests to any module or function you add. Also add doctests
Style guide
Ensure you run mix format to format the code to the best standards and limit failure when the CI runs.

          t() :: %ExUssd{
  data: map(),
  default_error: String.t(),
  delimiter: {String.t(), boolean()},
  error: {String.t(), boolean()},
  handle: term(),
  handler: (... -> any()),
  menu_list:
    {[
       %ExUssd{
         data: term(),
         default_error: term(),
         delimiter: term(),
         error: term(),
         handle: term(),
         handler: term(),
         menu_list: term(),
         name: term(),
         next: term(),
         orientation: term(),
         parent: term(),
         previous: term(),
         should_close: term(),
         show_navigation: term(),
         split: term(),
         title: term(),
         validation_menu: term()
       }
     ], boolean()},
  name: String.t(),
  next: {map(), boolean()},
  orientation: term(),
  parent: %ExUssd{
    data: term(),
    default_error: term(),
    delimiter: term(),
    error: term(),
    handle: term(),
    handler: term(),
    menu_list: term(),
    name: term(),
    next: term(),
    orientation: term(),
    parent: term(),
    previous: term(),
    should_close: term(),
    show_navigation: term(),
    split: term(),
    title: term(),
    validation_menu: term()
  },
  previous: {map(), boolean()},
  should_close: {boolean(), boolean()},
  show_navigation: {boolean(), boolean()},
  split: {integer(), boolean()},
  title: {String.t(), boolean()},
  validation_menu:
    {%ExUssd{
       data: term(),
       default_error: term(),
       delimiter: term(),
       error: term(),
       handle: term(),
       handler: term(),
       menu_list: term(),
       name: term(),
       next: term(),
       orientation: term(),
       parent: term(),
       previous: term(),
       should_close: term(),
       show_navigation: term(),
       split: term(),
       title: term(),
       validation_menu: term()
     }, boolean()}
}

  defmodule MyAppWeb.Router do
use Phoenix.Router
import ExUssd

scope "/", MyAppWeb do
  pipe_through [:browser]
  simulate "/simulator",
    menu: ExUssd.new(name: "Home", handler: MyHomeHandler),
    phone_numbers: ["254700100100", "254700200200", "254700300300"]
end
  end

defmodule MyHomeHandler do
  use ExUssd.Handler
  def init(menu, _api_parameters) do
    menu |> ExUssd.set(title: "Enter your pin number")
  end

  def callback(menu, api_parameters) do
    case api_parameters.text == "5555" do
      true ->
        menu
        |> ExUssd.set(title: "success, thank you.")
        |> ExUssd.set(should_close: true)

      _ ->
        menu |> ExUssd.set(error: "Wrong pin number\n")
    end
  end
end

defmodule MyHomeHandler do
  use ExUssd.Handler
  def init(menu, _api_parameters) do
    menu |> ExUssd.set(title: "Enter your pin number")
  end

  def callback(menu, api_parameters) do
    case api_parameters.text == "5555" do
      true ->
        menu
        |> ExUssd.set(title: "success, thank you.")
        |> ExUssd.set(should_close: true)

      _ ->
        menu |> ExUssd.set(error: "Wrong pin number\n")
    end
  end
end

	%{text: text, service_code: service_code}

  iex> ExUssd.Route.get_route(%{text: "544#", service_code: "544#"})
  [%{depth: 1, value: "555"}]
  iex> ExUssd.Route.get_route(%{text: "54423#", service_code: "544#"})
  [%{depth: 1, value: "3"}, %{depth: 1, value: "2"}, %{depth: 1, value: "555"}]
  iex> ExUssd.Route.get_route(%{text: "2", service_code: "*544#"})
  %{depth: 1, value: "2"}

